516-20
ADH
1/9/69

Card Reader Bootstrap and Loaders

Three loader programs have been written for use with the Soroban card reader through the I/O bus (not through the I/O loop). The first, called the Bootstrap Loader, is able to load cards punched in a special Bootstrap format (described below). The Bootstrap Loader resides in protected memory locations 1-178. The second loader, called the Binary Loader, loads cards punched in the absolute binary format described in document 516-16. The Binary Loader is punched in Bootstrap format on a single card, called the Bootstrap Card. The third loader, called the Patch Loader, loads absolute binary cards and octal correction (patch) cards. The Patch Loader is punched in absolute binary format; hence, it may be loaded by the Binary Loader.

The start-up or bootstrap loading of a deck goes as follows:

1) Turn on computer, but not card reader.

2) Hand load the Bootstrap Loader (see below) if necessary; normally this step may be skipped because memory protection should prevent destruction of the Bootstrap Loader.

3) Start computer running at location 1 (Master Clear, load 1 into P/Y, set switch to RUN, push START).

4) Place Bootstrap Card (containing Binary Loader) in front of absolute binary deck to be loaded. If any octal patching is required, put the Patch Loader deck between the Bootstrap Card and the absolute binary deck, and place the patch cards after the deck to be patched (but before the transfer card!).

5) Load the complete deck into the hopper and start the card reader.

6) If the computer halts before the complete deck has been read, the last card was in error, or it was read incorrectly. Take the last card read from the top of the output hopper, place it at the bottom of the input hopper, restart the card reader, and push START on the DDP-516 console.

Bootstrap Loader

The Bootstrap Loader occupies protected memory locations 1-178, as noted above. Hence, the Bootstrap Loader must be entered into

memory from the console. An octal listing of this Loader, with a symbolic equivalent for commentary, follows:

location	octal	symbolic
1	170340	SCCR
2	002001	JMP 1
3	131440	RCCR
4	002003	JMP 3
5	041464	LGL 12
6	130440	OCCR
7	002006	JMP 6
10	100001	SRC
11	050000	STA 0,1
12	024000	IRS 0
13	101001	SSC
14	010000	STA 0
15	100401	SRC∧SPL
16	002003	JMP 3
17	042000	JMP 0,1

Note that the index register, hence also location 0, is used by the Bootstrap Loader.

The Bootstrap card format places one data word in each pair of card columns. The top 7 rows of each odd-numbered column are ignored. The 8th row from the top (5 row) is a flag. The bottom 4 rows of the odd column and the entire following even column form a 16-bit word. If the flag is 1, the 16-bit word is data, to be stored in the next sequential memory location. If the flag is 0, the 16-bit word contains an address; the leading bit of the address specifies a new data loading origin if 1, or a jump address (loader escape) if 0. Pictorially:



It is useful to note that a jump to 0 causes the computer to halt; a jump to 1 causes the following card to be read.

Binary Loader (punched on Bootstrap Card)

The Binary Loader loads absolute binary cards. It is punched in Bootstrap format on a single card. The current version of this Loader occupies memory locations 177268-177778. A symbolic listing of the Binary Loader follows:

```
CNT     BSS     1
ADRES   BSS     1
CSUM    BSS     1
ILCNT   BSS     1
M3      DEC     -3
BFRA    ADDR    ADRES+1,1
DOIT            **
        CRA
        STA     CSUM
DLP     RCCR
        JMP     DLP
        IAB
        LDA     M3
        STA     ILCNT
DILP    RCCR
        JMP     DILP
        LGL     4
        LRR     4
        IMA     CSUM
        ADD     CSUM
        IRS     0
        JMP     PUTW
        SNZ
        JMP     DOIT,*
        HLT
START   SCCR
        JMP     START
        LDX     M3
        LDA     BFRA
        STA     ADRES
        JST     DOIT
        LDX     CNT
        IRS     CNT
        SKP
        JMP     ADRES,*
        JST     DOIT
        JMP     START
PUTW    IMA     CSUM
        STA     ADRES,*
        IRS     ILCNT
        JMP     DILP
        JMP     DLP
```

This Loader uses the X register (location 0). It is started at location START (177578 in the current version). If the Loader should halt (location 177568) due to a checksum error, it may be restarted simply by pushing the START button on the console.

Patch Loader (Binary loader with octal patch capability)

The Patch Loader loads absolute binary cards and octal patch cards. The current version of this loader is punched in absolute binary format; hence, it may be loaded by the Binary Loader. This Loader occupies core locations 175428-177308. The starting address for the Patch Loader is 176008. If the loader idles due to an error, it will resume automatically when the trouble has been cleared. If the error is not mechanical (card reader did not stop), then take the top card from the output stack, place it on the bottom of the input stack, and restart the card reader, to permit rereading of the card.

The format of an octal patch card follows:

```
Oaaaaa xxxxxx,...,xxxxxx
```

The "O" is punched in column 1, followed immediately by the octal address of the first octal data word. A single blank column separates the address from the first data word. Succeeding data words are separated by commas. The first blank in the data field terminates the scan. Leading zeroes may be omitted from address and data; a null entry (two adjacent commas) is treated as a zero word.

The Patch Loader also accepts an octal jump card, which causes control to pass immediately to the specified address:

```
Jaaaaa
```

The "J" is punched in column 1, followed immediately by the octal transfer address. Leading zeroes may be omitted; the first blank terminates the scan.

Segment Loader

The loader for segmented programs (see 516-14) has an octal patch capability. The data-generating card is an "O" card having exactly the same format as the "O" card for absolute programs (see 516-20-4). All "O" cards for a given segment must immediately follow the binary deck of that segment.

Sometimes it may be necessary to increase the size of an assembled segment, e.g., to make room for a patch which does not fit within the number of words which may be overwritten. In this case, an "S" (size) card may be placed immediately in front of the affected segment:



The "S" is punched in column 1, followed immediately by the size in octal (the highest address (RA) plus one). Leading zeroes may be omitted; the scan stops on the first blank.